

BIDC FIJI Automation Workshop

February 23, 2017

<http://imagej.net/developer/index.html>

Overall Goal

This workshop assumes some basic knowledge of FIJI (Fiji Is Just ImageJ) with the aim of acquainting researchers with the Macro Recorder utility, creating customized functions for more advanced users, and then employing FIJI's Batch Mode to analyze multiple images. The emphasis will be on creating *generalized* Macros that could work on any image, not just specific ones, at an introductory level.

The Macro Recorder

The Macro Recorder is an easy way to quickly set up the scaffolding for your Macro (often through trial and error of testing different processing functions within FIJI). The usefulness of the recorder is that it will record every action you take in FIJI which can easily be translated to a working script to apply to images. However, as we'll see, we won't need every single recorded action.

To start the Recorder:

Plugins -> Macros -> Record...

The Recorder window will appear. As you open an image and begin to manipulate it, you will notice your commands accumulating in this window.

Examples

As a trivial example, let's create a Macro that swaps the LookUpTable Colors between the blue and red channels for a 3-channel image.

1. Initialize the Recorder:

Plugins -> Macros -> Record...

2. Open the example fluorescent image in FIJI:

File -> Open Samples -> Fluorescent Cells (400k)

Notice that the Recorder window is no longer empty.

3. Now split the channels, and assign channel 3 to Red and channel 1 to Blue. Note that you'll need to activate the correct image by clicking on it, before changing the color:

```
Image -> Color -> Split Channels
Image -> Lookup Tables -> Red
Image -> Lookup Tables -> Blue
```

4. Now merge the channels to finish:

```
Image -> Color -> Merge Channels...
```

5. At the end, my recorder window looks like this:

```
run("Fluorescent Cells (400K)");
run("Split Channels");
selectWindow("C3-FluorescentCells-1.tif");
run("Red");
selectWindow("C1-FluorescentCells-1.tif");
run("Blue");
run("Merge Channels...", "c1=C1-FluorescentCells-1.tif
  c2=C2-FluorescentCells-1.tif
  c3=C3-FluorescentCells-1.tif create");
```

6. To make this a more generalized (able to apply to any 3-channel image), I'll change the recorder script to be the following. Note that all lines that begin with // are comments and are ignored by FIJI:

```
// select the active window and assign the image name to variable 't'
t = getTitle;

// split the channels, and swap the red with blue
run("Split Channels");

selectWindow("C3-" + t);

run("Red");

selectWindow("C1-" + t);

run("Blue");

// merge the channels back to make a composite image
// notice that we're taking advantage of string concatenation here
run("Merge Channels...", "c1=C1-" + t + " c2=C2-" + t + " c3=C3-" + t + " create");
```

User Defined Functions

Examples

Create a function that converts any 16-bit images to 8-bit, and add it to my channel-swapping Macro:

```
// select the active window and assign the image name to variable 't'
t = getTitle;

convert8bit();

// split the channels, and swap the red with blue
run("Split Channels");
selectWindow("C3-" + t);
run("Red");
selectWindow("C1-" + t);
run("Blue");

// merge the channels back to make a composite image
// notice that we're taking advantage of string concatenation here
run("Merge Channels...", "c1=C1-"+t+" c2=C2-"+t+" c3=C3-"+t+" create");

// User defined function

function convert8bit(){
    if (bitDepth==16){
        run("8-bit");
    }
}
```

Try this out on the HeLa Cells sample.

Putting it all Together with Batch Mode

Using FIJI's Batch mode addresses the idea of applying the Macro to a series of images.

Examples

Below is a basic skeleton to use. This code allows the user to choose the source folder location of the tif images to process, and the destination folder to save the processed images to. Your custom code can be pasted in the middle of the skeleton as indicated and user-defined functions should be pasted at the very end.

```

// BEGIN THE SKELETON RECIPE

// open dialog box to choose the source folder of images
sourcedir = getDirectory("Choose Source Directory ");

// open dialog box to the output images
destdir = getDirectory("Choose Destination Directory ");

// stores all files in the source directory as an array of strings
list = getFileList(sourcedir);

// enter Batch mode: images are not display and runs ~20 times faster
setBatchMode(true);

// loop through each image, applying whatever Macro you design and save the output
for (i=0; i<list.length; i++) {
    filename = sourcedir + list[i];

    // only choose input TIF files
    if (endsWith(filename, ".tif")) {
        open(filename);

        // YOUR MACRO BEGINS HERE

        // ENDS HERE

        // save images as TIF
        saveAs("TIFF", destdir+list[i]);
        close();
    }
}

// USER-DEFINED FUNCTIONS BEGIN HERE

```